So erstellen Sie automatisch eine Übersicht Ihrer Access-Objekte Teil 1
Versionen: Access 2003, 2002/XP, 2000 und 97

Sicherlich kennen Sie die Situation: Sie suchen eine bestimmte Abfrage für de Auswertung von Umsatzzahlen für den Vorgesetzten oder müssen für ein Meeting schnell einen Bericht ausdrucken. Aber in welcher der vielen Datenbanken befindet sich nun die Abfrage oder der Bericht? Während Sie nun Dutzende von Datenbanken vergeblich öffnen und durchsuchen, trommelt der Vorgesetzte ungeduldig mit den Fingern auf Ihrem Schreibtisch oder das Meeting läuft bereits eine Viertelstunde...

Hilfreich wäre hier eine Datenbank, in der Sie die Objekte Ihrer Datenbanken speichern und dann schnell beispielsweise nach einer Abfrage oder einem Bericht suchen könnten. Eine solche Lösung stellen wir Ihnen mit diesem zweiteiligen Tipp der Woche vor: Eine Prozedur liest die Objektinformationen (Name, Typ, Pfad/Dateiname der Datenbank) entweder von einer einzelnen Datenbank oder von allen in einem Verzeichnis gespeicherten Datenbanken in eine Tabelle "Objektübersicht" ein. Diese können Sie dann komfortabel durchsuchen.

Im ersten Teil des Tipps der Woche legen wir zunächst die benötigte Tabelle und die Prozedur zum Auslesen der Objektinfos an. Der zweite Teil erläutert dann die Funktionsweise und stellt Abfragen und Funktionen zum komfortablen Durchsuchen Ihrer Objektübersicht vor.

Für den Aufbau der Lösung legen Sie am besten zunächst eine neue, leere Datenbank "Objektübersicht und Suche" und darin eine neue Tabelle "Objektübersicht" mit den folgenden Feldern an:

Objektname: Datentyp "Text", Feldlänge = 100 Zeichen
Objekttyp: Datentyp "Text", Feldlänge = 50 Zeichen
Datenbank: Datentyp "Text", Feldlänge = 250 Zeichen

Erstellen Sie dann ein neues Modul "modObjects", setzen Sie über Extras-Verweise eine Referenz auf "Microsoft DAO X.x Object Library" (X.x = 3.6 für Access 2003, 2002/XP und 2000 und 3.5 für Access 97) und geben Sie anschließend die folgende Prozedur "GetObjects" ein bzw. kopieren Sie sie aus dem Newsletter in den VBA-Editor:

Sub GetObjects(strPath As String, _
               strMDB As String, _
               blnAppend As Boolean)

  Dim dbLocal As DAO.Database
  Dim dbExternal As DAO.Database
  Dim rs As DAO.Recordset
  Dim td As TableDef, qd As QueryDef
  Dim cont As Container, doc As Document
  Dim arrContData As Variant
  Dim I As Long, cntObjs As Long, cntDBs As Long
  Dim strFName As String, strDBName As String
  Dim strObjName As String, strObjType As String

  On Error Resume Next
  'Variablen initialisieren
  arrContData = Array("Forms", "Formular", _
                      "Reports", "Bericht", _
                      "Scripts", "Makro", _
                      "Modules", "Modul")

  If Not blnAppend Then 'Tabelleninhalt löschen
    DoCmd.SetWarnings False
    DoCmd.RunSQL "delete * from [Objektübersicht];"
    DoCmd.SetWarnings True
    DBEngine.Idle
  End If

  'Recordset initialisieren
  Set dbLocal = CurrentDb()
  Set rs = dbLocal.OpenRecordset("Objektübersicht")

  'Datenbank(en) scannen
  If strMDB <> "" Then 'Einzelne MDB
    strDBName = strMDB
    GoSub OneMDB
  Else 'MDBs in Verzeichnis
    If Right$(strPath, 1) <> "\" Then
      strPath = strPath & "\"
    End If
    strFName = Dir$(strPath & "*.mdb")
    While strFName <> ""
      strDBName = strPath & strFName
      GoSub OneMDB
      DoEvents
      strFName = Dir$() 'Nächste MDB...
    Wend
  End If 'Einzelne MDB/Verzeichnis?

  'Und fertig...
  rs.Close
  Set rs = Nothing
  Set dbLocal = Nothing
  Set dbExternal = Nothing

  DoEvents
  Beep
  MsgBox Format$(cntObjs, "###,###,###") & _
         " Objekt(e) aus " & _
         Format$(cntDBs, "###,###,###") & _
         " Datenbank(en) ausgelesen...", _
         vbOKOnly + vbInformation, "GetObjects:"
  Exit Sub '!!!

OneMDB:
  Err = 0
  Set dbExternal = OpenDatabase(strDBName)
  If Err <> 0 Then
    GoSub SaveErrorInfos
  Else
    'Tabellen und Abfragen über "Defs"-Auflistungen
    strObjType = "Tabelle"
    GoSub SaveTableInfos
    strObjType = "Abfrage"
    GoSub SaveQueryInfos
    'Rest über Container
    For I = 0 To UBound(arrContData) Step 2
      Set cont = dbExternal.Containers(arrContData(I))
      strObjType = arrContData(I + 1)
      GoSub SaveContInfos
    Next I
    dbExternal.Close
    cntDBs = cntDBs + 1
  End If 'Fehler?
  Return 'OneMDB

SaveErrorInfos:
  strObjName = Err.Description
  strObjType = CStr(Err.Number)
  GoSub SaveRecord
  Return 'SaveErrorInfos

SaveTableInfos:
  For Each td In dbExternal.TableDefs
    strObjName = td.Name
    If Left$(strObjName, 4) <> "MSys" Then
      GoSub SaveRecord
    End If
  Next td
  Return 'SaveTableInfos

SaveQueryInfos:
  For Each qd In dbExternal.QueryDefs
    strObjName = qd.Name
    If Left$(strObjName, 1) <> "~" Then
      GoSub SaveRecord
    End If
  Next qd
  Return 'SaveQueryInfos

SaveContInfos:
  For Each doc In cont.Documents
    strObjName = doc.Name
    GoSub SaveRecord
  Next doc
  Return 'SaveObjInfos

SaveRecord:
  rs.AddNew
  rs("Objektname") = strObjName
  rs("Objekttyp") = strObjType
  rs("Datenbank") = strDBName
  rs.Update
  cntObjs = cntObjs + 1
  Return 'SaveRecord

End Sub

Die Prozedur erwartet als ersten Parameter "strPath" die Angabe eines Verzeichnisses, in dem die Datenbanken gespeichert sind, deren Objektinfos Sie auslesen möchten:

GetObjects "D:\Daten\Access-Datenbanken", "", True

Wenn Sie diesen Parameter leer ("") angeben, können Sie über den zweiten Parameter "strMDB" auch eine einzelne Datenbank (Pfad/Dateiname) angeben, um nur daraus die Objektinfos auszulesen:

GetObjects "", "Z:\Test\Adressen.mdb", True

Der dritte Parameter "btnAppend" steuert, ob die Objektinfos zur Tabelle hinzugefügt werden sollen (= True) oder ob die Tabelle vorher gelöscht werden soll (= False). Letzteres wäre zum Beispiel notwendig, wenn Sie die Objektinfos aller Datenbanken aus einem Verzeichnis neu einlesen möchten:

GetObjects "F:\Verwaltung\Access-Datenbanken", "", False

Soviel zunächst für diesen Teil des Tipps der Woche. Im zweiten Teil in der nächsten Ausgabe erläutern wir die Funktion der Prozedur "GetObjects" und stellen Ihnen Abfragen und Funktionen zum komfortablen Durchsuchen der Objektübersicht vor.
So erstellen Sie automatisch eine Übersicht Ihrer Access-Objekte, Teil 2
Versionen: Access 2003, 2002/XP, 2000 und 97

Im ersten Teil dieses Tipps der Woche hatten wir Ihnen den Aufbau der Tabelle "Objektübersicht" und die Prozedur "GetObjects" für das Auslesen und Speichern von Objektinfos aus einer oder mehreren Datenbanken vorgestellt. Sehen wir nun zunächst an, wie die Prozedur "GetObjects" genau arbeitet.

Das Auslesen der Infos erfolgt auf zwei Arten: Die Informationen für Formulare, Berichte, Makros und Module lesen wir aus "Containern" aus, in denen Access diese Objekte als "Documents" speichert. Jedes "Document" gibt Auskunft über Name, Aufbau und weitere Eigenschaften des jeweiligen Objektes. Theoretisch lassen sich auf diesem Weg auch Infos zu Tabellen und Abfragen auslesen, Access fasst diese beiden Objekttypen aber in einem Container zusammen, so dass man zusätzlich den Objekttyp abfragen müsste, um nach Tabellen und Abfragen zu unterscheiden. Daher ist es einfacher, die Infos über die Auflistungen "TableDefs" und "QueryDefs" auszulesen. Das erfolgt über entsprechende, per "GoSub" angesprungene Unterroutinen "SaveTableInfos" und "SaveQueryInfos", während der Rest über eine Unterroutine "SaveContInfos" erledigt wird. Diese Unterroutinen rufen wiederum eine Unterroutine "SaveRecord" auf, die die Infos schließlich über einen Recordset in die Tabelle schreibt. Vor dem Aufruf der Unterroutinen setzen wir Objekttyp und Namen in die Variablen "strObjType" und "strObjName", während "strDBName" auf Pfad/Dateiname der jeweils aktuellen Datenbank gesetzt wird. Diese drei Informationen landen dann in den Feldern "Objektname", "Objekttyp" und "Datenbank" der Tabelle "Objektübersicht". Kommt es beim Zugriff auf eine Datenbank zu einem Fehler, wird die Unterroutine "SaveError" aufgerufen; sie schreibt Fehlerinformationen in die Tabelle "Objektübersicht", so dass Sie erkennen können, warum der Zugriff auf eine Datenbank gescheitert ist.

Um nun eine bestimmte Abfrage, einen bestimmten Bericht oder irgendein anderes Objekt im Bestand Ihrer Datenbanken zu suchen, ist der einfachste Weg eine Parameterabfrage:

1. Wechseln Sie im Datenbankfenster auf die Registerkarte "Abfragen" und klicken Sie auf Neu.

2. Wählen Sie "Entwurfsansicht" und dann die Tabelle "Objektübersicht" aus und klicken Sie auf Schließen.

3. Fügen Sie alle drei Felder per Doppelklick in die Feldliste zum Abfrageentwurf hinzu.

4. Wählen Sie das Menü Abfrage-Parameter an, richten Sie die folgenden Parameter ein und klicken Sie auf Ok:

Typ des Objektes: Text
Name des Objektes: Text
· Klicken Sie im Abfrageentwurf in die Zeile "Kriterien" des Feldes "Objektname" und geben Sie folgenden Ausdruck ein:

Wie [Name des Objektes]
· Klicken Sie im Abfrageentwurf in die Zeile "Kriterien" des Feldes "Objekttyp" und geben Sie folgenden Ausdruck ein:

Wie [Typ des Objektes]
· Speichern Sie die Abfrage beispielsweise unter "Objekte suchen" und lassen Sie sie einmal testweise ausführen.

Es wird nun zunächst der Dialog "Parameterwerte eingeben" für beide Parameter angezeigt. Geben Sie hier zum Beispiel "Tabelle" und "Kunden" ein, um alle Datenbanken auflisten zu lassen, in denen eine solche Tabelle vorhanden ist. Für "Typ des Objektes" können Sie "Tabelle", "Abfrage", "Formular", "Bericht", "Makro" oder "Modul" bzw. ein Sternchen "*" eingeben, wenn der Typ des Objektes keine Rolle spielt. Um also alle Objekte mit dem Namen "Kunden" aufzulisten, geben Sie für "Typ des Objektes" ein Sternchen und für "Name des Objektes" den Begriff "Kunden" ein. Umgekehrt funktioniert es natürlich auch: Wenn Sie alle Tabellen, egal welchen Namen diese haben, auflisten möchten, geben Sie als "Typ des Objektes" den Begriff "Tabelle" und für "Name des Objektes" ein Sternchen "*" ein.

Ähnlich einfach lösen Sie folgendes Problem: Sie wissen zwar den Namen einer Datenbank, aber deren Speicherort nicht mehr. Legen Sie wie oben erläutert eine Abfrage "Datenbank suchen" an, richten Sie einen Parameter "Name der DB" ein und tragen Sie in der Zeile "Kriterien" des Feldes "Datenbank" den folgenden Ausdruck ein:

Wie "*" & [Name der DB] & "*"

Wenn Sie nun beispielsweise "Adr" eingeben, werden alle Datenbanken mit "Adr" im Namen wie zum Beispiel "Adressen.mdb", "Adressverwaltung.mdb" oder "Kundenadressen.mdb" aufgelistet.

